
Abstract. This paper describes the RationalMaps package for Macaulay2. This package provides
functionality for computing several aspects of rational maps.

1. Introduction

This package aims to compute a number of things about rational maps between varieties. In
particular, this package will compute

◦ The base locus of a rational map.
◦ Whether a rational map is birational.
◦ The inverse of a birational map.
◦ Whether a map is a closed embedding.
◦ And more!

Our functions have numerous options which allow them to run much more quickly in certain ex-
amples if configured correctly. The Verbose option gives hints as to the best way to apply these.

Suppose F : X −→ Y is given by homogeneous forms f of the same degree. The idea of looking at
the syzygies of the forms f to detect the geometric properties of F goes back at least to [HKS92] in
the case where X = Pn, Y = Pm and m = n (see also [ST69]). In [RS01] this method was developed
by Russo and Simis to handle the case X = Pn and m ≥ n. Simis pushed the method further to the
study of general rational maps between two integral projective schemes in arbitrary characteristic
by an extended ideal-theoretic method emphasizing the role of the Rees algebra associated to the
ideal generated by f [Sim04]. Recently, Doria, Hassanzadeh, and Simis applied these Rees algebra
techniques to study the birationality of F [DHS12]. Our core functions, in particular the functions
related to computing inverse maps, rely heavily on this work.

2. Base Loci

We begin with the problem of computing the base locus of a map to projective space. Let X be
a projective variety over any field k and let F : X −→ Pm

k be a rational map from X to projective

space. Then we can choose some representative (f0, · · · , fm) of F, where each fi is the ith coordinate
of F. A priori, each fi is in K = fracR, where R is the coordinate ring of X. However, we can get
another representative of F by clearing denominators. (Note this does not enlarge the base locus of

F since F is undefined whenever the denominator of any of the fi vanishes.) Thus we can assume
that fi ∈ R for all i, and that all the fi are homogenous of the same degree.

In this setting, one might naively think that the map F is undefined exactly when all of the fi
vanish, and thus the base locus is the vanishing set of the ideal (f0, · · · , fm). However, this can
yield a base locus that’s too big. Indeed, to find the base locus of a rational map, we must consider
all possible representatives of the map and find where none of them are defined. To do this, we use
the following result.

Proposition 2.1. [Sim04, Proposition 1.1] Let F : X 99K Pm be a rational map and let f =
{f0, . . . , fm} be a representative of F with fi ∈ R homogenous of degree d for all i. Set I =
(f0, · · · , fm). Then the set of such representatives of F corresponds bijectively to the homogenous
vectors in the rank 1 graded R-module HomR(I,R) ∼= (R :K I).

The bijection comes from multiplying our fixed representative f of F by h ∈ (R :K I). Now, in
the setting of Proposition 2.1, let⊕

s

R(ds)
ϕ−→ R(−d)m

[f0,··· ,fm]−−−−−−→ I −→ 0

be a free resolution of I. Then we get

0 −→ HomR(I,R) −→
(
R(−d)m+1

)∨ ϕt

−→

(⊕
s

R(ds)

)∨
where ϕt is the transpose of ϕ and R∨ is the dual module of R. Thus, we get that HomR(I,R) ∼=
kerϕt, and so each representative of F corresponds to a vector in kerϕt. The correspondence takes
a representative (hf0, · · · , hfm) to the map that multiplies vectors in Rm+1 by [hf0, · · · , hfm] on
the left.

The base locus of F is the intersection of the sets V (f i0, · · · , f im) as f i = (f i0, · · · , f im) ranges over
all the representatives of F. The above implies that this is the same as the intersection of the sets
V (wi

0, · · · , wi
m) as wi = (wi

0, · · · , wi
m) ranges over the vectors in kerϕt. Now, given any a, f, g ∈ R,

we have V (af) ⊇ V (f) and V (f + g) ⊇ V (f) ∩ V (g). Thus, it’s enough to take a generating set
w1, · · · ,wn of kerϕt and take the intersection over this generating set.

The base locus of F is then the variety cut out by the ideal generated by all the entries of all of
the wi. Our function baseLocusOfMap returns this ideal.
i1 : loadPackage "RationalMaps";

i2 : R = QQ[x,y,z];

i3 : f = {x^2*y, x^2*z, x*y*z};

i4 : baseLocusOfMap(f);

o4 = ideal (y*z, x*z, x*y)

o4 : Ideal of R

If the SaturateOutput option is set true, our function will return the saturation of this ideal.

3. Birationality and Inverse Maps

Again, a rational map F : X ⊆ Pn 99K Y ⊆ Pm between projective spaces is defined bym+1 forms
f = {f0, . . . fm} of the same degree in n+ 1 variables, not all vanishing. Let f = {f0, . . . , fm} ⊂ R
be an ordered set of forms of the same degree, where R, the coordinate ring of X, denotes a
standard graded k-algebra, with k a field and dimR ≥ 1. Thus, one has a presentation R '
k[x0, . . . , xn] = k[X]/a with k[X] = k[X0, . . . , Xn] standard graded and a a homogeneous ideal. Let
R[Y] = R[Y0, . . . , Ym] be a polynomial ring over R with the standard bigrading where deg(Xi) =
(1, 0) and deg(Yj) = (0, 1). Since I = (f0, . . . , fm) is generated in a fixed degree, the Rees algebra

RR((f)) := R⊕ I ⊕ I2 ⊕ · · · ' R[It] ⊂ R[t]

is a standard bigraded k-algebra. Mapping Yj 7→ fjt yields a presentation R[Y]/J ' RR((f)), with
J a bihomogeneous presentation ideal. J depends only on the rational map defined by f and not on
this particular representative. Write

J =
⊕

(p,q)∈N2

J(p,q),

where J(p,q) denotes the k-vector space of forms of bidegree (p, q).
For birationality, the following bihomogeneous piece is important:

J1,∗ :=
⊕
r∈N

J1,q

with J1,q denoting the bigraded piece of J spanned by the forms of bidegree (1, q) for all q ≥ 0.
Now, a form of bidegree (1, ∗) can be written as

∑n
i=0Qi(Y)xi, for suitable homogeneous Qi(Y) ∈

k[Y] ⊂ R[Y] of the same degree. Since the Y are indeterminates over R, each pair of such
representations of the same form gives a syzygy of {x0, . . . , xn} with coefficients in k. Thus every
such representation is unique up to k-linear relations of {x0, . . . , xn}, i.e., up to elements of a1, the
degree-1 part of a.

Next, one can pick a minimal set of generators of the ideal (J1,∗) consisting of a finite number of
forms of bidegree (1, q), for various q’s. Let {P1, . . . , Ps} ⊂ k[X,Y] denote liftings of these biforms
and let {`1, . . . , `r} ⊂ k[X] be a k-vector space basis of a1. Consider the Jacobian matrix of the
polynomials {`1, . . . , `r, P1, . . . , Ps} with respect to X. This is a matrix with entries in k[Y]. Write
ψ for the corresponding matrix over S = k[Y]/b, the coordinate ring of Y . This matrix is called
called the weak Jacobian dual matrix associated to the given set of generators of (J1,∗). Note that
a weak Jacobian matrix ψ is not uniquely defined due to the lack of uniqueness in the expression
of an individual form and to the choice of bihomogeneous generators. However, it is shown in
[DHS12, Lemma 2.13] that if the weak Jacobian matrix associated to one set of bihomogeneous
minimal generators of (J1,∗) has (finite) rank over S then the weak Jacobian matrix associated to
any other set of bihomogeneous minimal generators of (J1,∗) has (finite) rank over S and the two
ranks coincide.

The main theorem of loc.cit. is the following:

Theorem 3.1. Let X ⊆ Pn be non-degenerate. Then F is birational if and only if rank(ψ) =
edim(R)− 1(= n). Moreover

(i) We get a representative for the inverse of F by taking the coordinates of any homogeneous
vector of positive degree in the (rank one) null space of ψ over S for which these coordinates
generate an ideal containing a regular element.

(ii) If, further, R is a domain, the representative of F in (i) can be taken to be the set of the
(ordered, signed) (edim(R)− 1)-minors of an arbitrary (edim(R)− 1)× edim(R) submatrix
of ψ of rank edim(R)− 1.

As expected, the most expensive part of applying this theorem is computing the Rees ideal J. In
the package RationalMaps we use ReesStrategy to compute the Rees equations. The algorithm
is the standard elimination technique. However we do not use the ReesAlgebra package, since
verifying birationality according to Theorem 3.1 only requires to computing a small part of the
Rees ideal, namely elements of first-degree 1. This idea is applied in the SimisStrategy. More
precisely, if the given map F is birational, then the jacobian dual rank will attain its maximum
value of edim(R)−1 after computing the Rees equations up to degree (1, N) for N sufficiently large.
This allows us to compute the inverse map. The down-side of SimisStrategy is that if F is not
birational, the desired number N cannot be found and the process never terminates. To provide

a definitive answer for birationality, we use HybridStrategy, which is a hybrid of ReesStrategy
and SimisStrategy. The default strategy is HybridStrategy.

HybridLimit is an option to switch SimisStrategy to ReesStrategy, if the computations up to
degree (1, HybridLimit) do not lead to rank(ψ) = edim(R)−1. The default value for HybridLimit
is 15. The change from SimisStrategy to ReesStrategy is done in such a way that the generators
of the Rees ideal computed in the SimisStrategy phase are not lost; the program computes other
generators of the Rees ideal while keeping the generators it found before attaining HybridLimit.

There is yet another method for computing the Rees ideal called SaturationStrategy. In this
option the whole Rees ideal is computed by saturating the defining ideal of the symmetric algebra
with respect to a non-zero element in R (we assume R to be a domain). This Strategy appears to
be slower in some examples, though one might be able improve this option in the future by stopping
the computation of the saturation at a certain step.

Computing inverse maps is the most important functionality of this package, and is done by the
function inverseOfMap. According to Theorem 3.1, there are two ways to compute the inverse of a
map: (1) by finding any syzygy of the jacobian dual matrix, and (2) by finding a sub-matrix of ψ of
rank edim(R)− 1. Each way has its own benefits. Method (1) is quite fast in many cases, however
method (2) is very useful if the rank of the jacobian dual matrix ψ is relatively small compared to
the degrees of the entries of ψ. Our function inverseOfMap starts by using the second method and
later switches to the first method if the second method didn’t work. The timing of this transition
from the first method to the second method is controlled by the option MinorsCount. Setting
MinorsCount to zero will mean that no minors are checked and the inverse map is computed just
by looking at the syzygies of ψ. If MinorsCount is left as null (the default value), the program will
try to make an educated guess as to how big to set this option, depending on varieties the user is
working with.

In addition, to improve the speed of the function inverseOfMap, we have two other options,
AssumeDominant and CheckBirational. If AssumeDominant is set to be true, then inverseOfMap

assumes that the map from X to Y is dominant and does not compute the image of the map; this
is time consuming in certain cases. Similarly, if CheckBirational set false, inverseOfMap will
not check birationality although it still computes the jacobian dual matrix.

In general, as long as Verbose is true, the function will make suggestions as to how to run it
more quickly. For example.

i1 : loadPackage "RationalMaps";

i2 : Q=QQ[x,y,z,t,u];

i3 : phi=map(Q,Q,matrix{{x^5,y*x^4,z*x^4+y^5,t*x^4+z^5,u*x^4+t^5}});

o3 : RingMap Q <--- Q

i4 : time inverseOfMap(phi, AssumeDominant=>true,CheckBirational=>false, MinorsCount=>50000)

Starting inverseOfMapSimis(SimisStrategy or HybridStrategy)

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 1}.

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 2}.

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 4}.

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 7}.

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 11}.

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 16}.

inverseOfMapSimis: We give up. Using all of the previous computations,

we compute the whole Groebner basis of the rees ideal.

Increase HybridLimit and rerun to avoid this.

inverseOfMapSimis: Found Jacobian dual matrix (or a weak form of it), it has 5 columns

and about 20 rows.

inverseOfMapSimis: Looking for a nonzero minor

Starting nonZeroMinor, looking for rank: 4, we will run it 50000 times.

If this is slow, rerun with MinorsCount=>0.

nonZeroMinor: Found a nonzero minor

inverseOfMapSimis: We found a nonzero minor.

-- used 3.51586 seconds

4. Embeddings

Our package also checks whether a rational map F : X −→ Y is a closed embedding. The strategy
is quite simple.

(a) We first check whether F is regular (by checking if its base locus is empty).
(b) We next invert the map (if possible).
(c) Finally, we check if the inverse map is also regular.

If all three conditions are met, then the map is a closed embedding and the function returns true.
Otherwise isEmbedding returns false.

i1 : needspackage "divisor"; --used to quickly define a map

i2 : c = QQ[x,y,z]/(x^4+x^2*y*z+y^4+z^3*x);

i3 : q = ideal(y,x+z); --a point on our curve

i6 : loadpackage "rationalmaps";

i8 : f2 = maptoprojectivespace(12*divisor(q));

o8 : RingMap C <--- QQ[YY , YY , YY , YY , YY , YY , YY , YY , YY , YY ]

1 2 3 4 5 6 7 8 9 10

i9 : time isEmbedding(f2)

isEmbedding: About to find the image of the map.

If you know the image, you may want to specify that and set AssumeDominant=>true option if

this is slow.

isEmbedding: Checking to see if the map is a regular map

isEmbedding: computing the inverse map

Starting inverseOfMapSimis(SimisStrategy or HybridStrategy)

inverseOfMapSimis: About to compute partial Groebner basis of rees ideal up to degree {1, 1}.

inverseOfMapSimis: We computed enough of the Groebner basis.

inverseOfMapSimis: Found Jacobian dual matrix (or a weak form of it), it has 3 columns and

about 17 rows.

inverseOfMapSimis: Failed to find a nonzero minor. We now compute syzygies instead.

If this doesn’t terminate quickly, you may want to try increasing the option MinorsCount.

isEmbedding: checking if the inverse map is a regular map

-- used 3.01948 seconds

o9 = true

5. Functionality overlap with other packages

We note that our package has some overlaps in functionality with other packages.
While the Parametrization package [Boe10] focuses mostly on curves, it also includes a function

called invertBirationalMap which has the same functionality as inverseOfMap. On the other
hand, these two functions were implemented somewhat differently and so sometimes one function
can be substantially faster than the other.

The package Cremona [Sta16] focuses on very fast probabilistic computation in general cases and
very fast deterministic computation for special kinds of maps from projective space. In particular,
in Cremona,

◦ isBirational gives a probabilisitc answer to the question of whether a map between vari-
eties is birational. Furthermore, if the source is projective space, then degreeOfRationalMap

with MathMode=>true can give a deterministic answer that is frequently faster than what
our package can provide with isBirationalMap.
◦ invertBirMap gives a very fast computation of the inverse of a birational map if the source

is projective space and the map has maximal linear rank. If you pass this function a map
not from projective space, then it calls a modified, faster version of invertBirationalMap
originally from Parametrization. Even in some cases with maximal linear rank, our
inverseOfMap function appears to be quite competitive however.

5.1. Speed comparisons. First we do comparisons using examples with maximal linear rank
where Cremona excels. These examples were run using version 2.0 of Cremona and version 0.21 of
RationalMaps and using version 1.9.1 of Macaulay2.

Indeed, in some of those examples with maximal linear rank, Cremona is substantially faster.

i1 : loadPackage "Cremona"; loadPackage "RationalMaps";

i3 : ringP20=QQ[t_0..t_20];

i4 : phi=map(ringP20,ringP20,{t_10*t_15-t_9*t_16+t_6*t_20,t_10*t_14-t_8*t_16+t_5*t_20,t_9*t_14-t_8*t_15+t_4*t_20,

t_6*t_14-t_5*t_15+t_4*t_16,t_11*t_13-t_16*t_17+t_15*t_18-t_14*t_19+t_12*t_20,t_3*t_13-t_10*t_17+t_9*t_18-t_8*t_19

+t_7*t_20,t_10*t_12-t_2*t_13-t_7*t_16-t_6*t_18+t_5*t_19,t_9*t_12-t_1*t_13-t_7*t_15-t_6*t_17+t_4*t_19,t_8*t_12

-t_0*t_13-t_7*t_14-t_5*t_17+t_4*t_18,t_10*t_11-t_3*t_16+t_2*t_20,t_9*t_11-t_3*t_15+t_1*t_20,t_8*t_11-t_3*t_14

+t_0*t_20,t_7*t_11-t_3*t_12+t_2*t_17-t_1*t_18+t_0*t_19,t_6*t_11-t_2*t_15+t_1*t_16,t_5*t_11-t_2*t_14+t_0*t_16,

t_4*t_11-t_1*t_14+t_0*t_15,t_6*t_8-t_5*t_9+t_4*t_10,t_3*t_6-t_2*t_9+t_1*t_10,t_3*t_5-t_2*t_8+t_0*t_10,t_3*t_4

-t_1*t_8+t_0*t_9,t_2*t_4-t_1*t_5+t_0*t_6});

i5 : time inverseOfMap(phi, AssumeDominant=>true, Verbose=>false)-- Function from "RationalMaps"

-- used 1.12522 seconds

i6 : time invertBirMap phi -- Function from "Cremona"

-- used 0.0789913 seconds

i7 : isSameMap(o5, o6) -- Function from "RationalMaps"

o7 = true

However, sometimes our function is substantially faster even in examples with maximal linear
rank.

i1 : ZZ/33331[t_0..t_6];

i2 : loadPackage "Cremona"; loadPackage "RationalMaps";

i4 : phi=toMap minors(3,matrix{{t_0..t_4},{t_1..t_5},{t_2..t_6}});

i5 : J=kernelComponent(phi,2);

i6 : phi=toMap(phi,Dominant=>J);

i7 : time inverseOfMap(phi, AssumeDominant=>true, Verbose=>false) -- Function from "RationalMaps"

-- used 0.126314 seconds

i8 : time invertBirMap phi; -- Function from "Cremona"

-- used 0.375856 seconds

i9 : isSameMap(o7, o8)

o9 = true

Note both these examples were taken directly from the documentation of Cremona. On the other
hand, when the source and target are not projective space, or the map does not have maximal
linear rank, in our experience, our functions appear to be faster. Here is an example where the
source and target are not projective space.

i1 : needsPackage "Divisor"; loadPackage "RationalMaps"; loadPackage("Cremona", Reload=>true);

i5 : C = QQ[x,y,z]/(x^4+x^2*y*z+y^4+z^3*x);

i6 : Q = ideal(y,x+z); --a point on our curve

i7 : (f3 = mapOntoImage(mapToProjectiveSpace(12*divisor(Q))););

i8 : (time g1 = inverseOfMap(f3, Verbose=>false);); -- Function from "RationalMaps"

-- used 1.1996 seconds

i9 : (time g2 = invertBirMap(f3);); -- Function from "Cremona"

-- used 13.7981 seconds

i10 : (g3 = map(target g1, source g2, sub(matrix g2, target g1)););

i11 : isSameMap(g3, g1)

o11 = true

We also include an example where the map does not have maximal linear rank.

i1 : loadPackage "RationalMaps"; loadPackage "Cremona";

i3 : Q=QQ[x,y,z,t,u];

i4 : phi=map(Q,Q,matrix{{x^5,y*x^4,z*x^4+y^5,t*x^4+z^5,u*x^4+t^5}});

o4 : RingMap Q <--- Q

i5 : (time inverseOfMap(phi, AssumeDominant=>true,CheckBirational=>false, MinorsCount=>50000));

-- Function from "RationalMaps"

-- used 5.80638 seconds

i9 : (time h = invertBirMap(phi)); -- Function from "Cremona"

-- used 71.8998 seconds

i10 : isSameMap(o5, h)

o10 = true

